
1. Optimization

minimize f (x)
subject to b ≤ h(x) ≤ b + r,

l ≤ x ≤ u

• Linear Programming (LP): f and h are linear.

• Convex Optimization: f is convex, each hi is concave, and r = ∞.

• Nonlinear Optimization: f and each hi is assumed to be twice
differentiable

• Generally, we seek a local solution in the vicinity of a given starting
point.

• If problem is convex (which includes LP), any local solution is au-
tomatically a global solution.

2. Least Action Principle

Given: n bodies.

Let:
mj denote the mass and
zj(t) denote the position in R2 = C of body j at time t.

Action Functional:

A =

∫ 2π

0

∑
j

mj

2
‖żj‖2 +

∑
j,k:k<j

mjmk

‖zj − zk‖

 dt.

3. Equation of Motion

First Variation:

δA =

∫ 2π

0

∑
α

∑
j

mjż
α
j δ̇z

α

j −
∑

j,k:k<j

mjmk

(zα
j − zα

k )(δzα
j − δzα

k )

‖zj − zk‖3

 dt

= −
∫ 2π

0

∑
j

∑
α

mjz̈
α
j +

∑
k:k 6=j

mjmk

zα
j − zα

k

‖zj − zk‖3

 δzα
j dt

Setting first variation to zero, we get:

mjz̈
α
j = −

∑
k:k 6=j

mjmk

zα
j − zα

k

‖zj − zk‖3
, j = 1, 2, . . . , n, α = 1, 2

Note: If mj = 0 for some j, then the first order optimality condition
reduces to 0 = 0, which is not the equation of motion for a massless
body.

4. Periodic Solutions

We assume solutions can be expressed in the form

zj(t) =

∞∑
k=−∞

γke
ikt, γk ∈ C.

Writing with components zj(t) = (xj(t), yj(t)) and γk = (αk, βk), we
get

x(t) = a0 +

∞∑
k=1

(
ac

k cos(kt) + as
k sin(kt)

)
y(t) = b0 +

∞∑
k=1

(
bc
k cos(kt) + bs

k sin(kt)
)

where

a0 = α0, ac
k = αk + α−k, as

k = β−k − βk,

b0 = β0, bc
k = βk + β−k, bs

k = αk − α−k.

The variables a0, ac
k, as

k, b0, bc
k, and bs

k are the decision variables in the
optimization model.

5. The ampl Model

param N := 3; # number of masses
param n := 15; # number of terms in Fourier series representation
param m := 100; # number of terms in numerical approx to integral

param theta {j in 0..m-1} := j*2*pi/m;

param a0 {i in 0..N-1} default 0; param b0 {i in 0..N-1} default 0;
var as {i in 0..N-1, k in 1..n} := 0; var bs {i in 0..N-1, k in 1..n} := 0;
var ac {i in 0..N-1, k in 1..n} := 0; var bc {i in 0..N-1, k in 1..n} := 0;

var x {i in 0..N-1, j in 0..m-1}
= a0[i]+sum {k in 1..n} ( as[i,k]*sin(k*theta[j]) + ac[i,k]*cos(k*theta[j]) );

var y {i in 0..N-1, j in 0..m-1}
= b0[i]+sum {k in 1..n} ( bs[i,k]*sin(k*theta[j]) + bc[i,k]*cos(k*theta[j]) );

var xdot {i in 0..N-1, j in 0..m-1}
= if (j<m-1) then (x[i,j+1]-x[i,j])*m/(2*pi) else (x[i,0]-x[i,m-1])*m/(2*pi);

var ydot {i in 0..N-1, j in 0..m-1}
= if (j<m-1) then (y[i,j+1]-y[i,j])*m/(2*pi) else (y[i,0]-y[i,m-1])*m/(2*pi);

var K {j in 0..m-1} = 0.5*sum {i in 0..N-1} (xdot[i,j]^2 + ydot[i,j]^2);

var P {j in 0..m-1}
= - sum {i in 0..N-1, ii in 0..N-1: ii>i}

1/sqrt((x[i,j]-x[ii,j])^2 + (y[i,j]-y[ii,j])^2);

minimize A: (2*pi/m)*sum {j in 0..m-1} (K[j] - P[j]);

http://www.princeton.edu/~rvdb

Continued...

let {i in 0..N-1, k in 1..n} as[i,k] := 1*(Uniform01()-0.5);
let {i in 0..N-1, k in 1..n} ac[i,k] := 1*(Uniform01()-0.5);
let {i in 0..N-1, k in n..n} bs[i,k] := 0.01*(Uniform01()-0.5);
let {i in 0..N-1, k in n..n} bc[i,k] := 0.01*(Uniform01()-0.5);

solve;

6. Choreographies and the Ducati

The previous ampl model was used to find many choreographies (a la
Moore and Montgomery/Chencinier) in the equimass n-body problem
and the stable Ducati solution to the 3-body problem.

